Vakiokertoimisen lineaariyhtälön karakteristisen yhtälön useampikertaiset juuret

Kiinnitetään aluksi kolme vakiota.

Tarkastelun kohteena olkoon kertalukua n oleva vakiokertoiminen homogeeninen differentiaaliyhtälö:

```
> n:= 3;
```

$$n := 3$$

Tämän karakteristinen polynomi on myös astetta n. Polynomilla olkoon nollakohta r, jonka kertaluku on p (<= n):

```
> p:= 2;
```

$$p := 2$$

Tarkoituksena on tutkia, onko $y = x^k\,\mathbf{e}^{rx}$ yhtälön ratkaisu, kun k = 0, 1, 2, ..., q (luontevimmin q >= p):

```
> q:= 3;
```

$$q := 3$$

Vakiokertoimisten yhtälöiden teorian mukaan näin pitäisi olla arvoon $k = p$ - 1 saakka.

Differentiaaliyhtälö on

```
> diffyht:= sum(a[k]*diff(y(x), [x$k]), k=0..n)=0;
```

$$diffyht := a_0\,\mathrm{y}(x) + a_1\left(\frac{d}{dx}\,\mathrm{y}(x)\right) + a_2\left(\frac{d^2}{dx^2}\,\mathrm{y}(x)\right) + a_3\left(\frac{d^3}{dx^3}\,\mathrm{y}(x)\right) = 0$$

Karakteristinen yhtälö saadaan sijoittamalla yhtälöön yrite $y = \mathbf{e}^{rx}$ ja jakamalla sijoittamisen jälkeen eksponenttitekijä pois. Samaan tulokseen päästään korvaamalla differentiaaliyhtälössä derivaatat vastaavilla muuttujan r potensseilla:

```
> seq(diff(y(x), [x$k])=r^k, k=0..n):
  karaktyht:= subs({%}, diffyht);
```

$$karaktyht := a_0 + a_1\,r + a_2\,r^2 + a_3\,r^3 = 0$$

Luku r on p-kertainen juuri, jos ja vain jos se on myös derivaattojen nollakohta kertalukuun p - 1 saakka. Kertaluku voidaan siten karakterisoida seuraavilla ehdoilla:

```
> ehdot:= seq(diff(karaktyht, [r$k]), k=0..p-1);
```

$$ehdot := a_0 + a_1\, r + a_2\, r^2 + a_3\, r^3 = 0,\ a_1 + 2\, a_2\, r + 3\, a_3\, r^2 = 0$$

Sijoitetaan ratkaisut $y = x^k\, \mathbf{e}^{rx}$ (0 <= k <= q) differentiaaliyhtälöön ja tutkitaan, toteutuuko tämä, kun r täyttää edellä asetetut ehdot:

```
> subs(y(x)=x^k*exp(r*x), diffyht):
  sij:= seq(%, k=0..q):
  simplify({%});
```

$$\{\mathbf{e}^{(r\,x)}\,(a_0\, x^3 + 3\, a_1\, x^2 + a_1\, x^3\, r + 6\, a_2\, x + 6\, a_2\, x^2\, r + a_2\, x^3\, r^2 + 6\, a_3 + 18\, a_3\, x\, r + 9\, a_3\, x^2\, r^2 + a_3\, x^3\, r^3) = 0,\ \mathbf{e}^{(r\,x)}\,(a_0\, x^2 + 2\, a_1\, x + a_1\, x^2\, r + 2\, a_2 + 4\, a_2\, x\, r + a_2\, x^2\, r^2 + 6\, a_3\, r + 6\, a_3\, x\, r^2 + a_3\, x^2\, r^3) = 0,\ \mathbf{e}^{(r\,x)}\,(a_0\, x + a_1 + a_1\, x\, r + 2\, a_2\, r + a_2\, x\, r^2 + 3\, a_3\, r^2 + a_3\, x\, r^3) = 0,\ \mathbf{e}^{(r\,x)}\,(a_0 + a_1\, r + a_2\, r^2 + a_3\, r^3) = 0\}$$

```
> simplify({sij}, {ehdot});
```

$$\{6\, a_2\, x\, \mathbf{e}^{(r\,x)} + 6\, a_3\, \mathbf{e}^{(r\,x)} + 18\, a_3\, x\, r\, \mathbf{e}^{(r\,x)} = 0,\ 2\, a_2\, \mathbf{e}^{(r\,x)} + 6\, a_3\, r\, \mathbf{e}^{(r\,x)} = 0,\ 0 = 0\}$$

```
>
```

Yhtälön toteuttavia ratkaisuja näyttää todellakin olevan nollakohdan r kertaluvun p mukainen määrä, kuten teorian mukaan pitääkin.

Polynomiehtojen käsittelyyn symboliset laskentajärjestelmät kuten *Maple* käyttävät ns. *Gröbnerin kantoja*. Probleeman monimutkaistuessa näiden käyttö tulee kuitenkin raskaaksi hyvin nopeasti. Vaikka edellä oleva lasku voidankin periaatteessa laskea millä tahansa arvoilla n, p, q laskenta alkaa arvojen kasvaessa vaatia aikaa ja muistia todella paljon. Lukija kokeilkoon!

Linkit

vakiokertoiminen homogeeniyhtälö

SKK & MS 31.05.2001